LA-2329
c.3

LOS ALAMOS SCIENTIFIC LABORATORY
OF THE UNIVERSITY OF CALIFORNIA o LOS ALAMOS NEW MEXICO

PUQFUA: AN IBM-704 FORTRAN CODE FOR DETERMINING PLUTONIUM BODY BURDEN FROM URINE ASSAYS

Printed in USA. Price $ 1.00. Available from the

Office of Technical Services
U. S. Department of Commerce
Washington 25, D. C.

LA-2329
BIOLOGY AND MEDICINE
TID-4500, 15th Ed.

LOS ALAMOS SCIENTIFIC LABORATORY

OF THE UNIVERSITY OF CALIFORNIA LOS ALAMOS NEW MEXICO

REPORT WRITTEN: July 1959

REPORT DISTRIBUTED: April 25, 1960

PUQFUA: AN IBM-704 FORTRAN CODE FOR DETERMINING PLUTONIUM BODY BURDEN FROM URINE ASSAYS

by

James N. P. Lawrence

Contract W-7405-ENG. 36 with the U. S. Atomic Energy Commission

ABSTRACT

PUQFUA is an IBM-704 FORTRAN program, written primarily for Pu^{239}, which calculates from urinalyses the plutonium body burden in microcuries and the per cent of the maximum permissible body burden. It is based on a set of power function elimination equations developed by Wright H. Langham for the excretion of plutonium over a five year period. Results with PUQFUA compare favorably with those of other methods. In general, it is easy to apply for a large number of individual samples, and the validation method is objective and requires no past experience with plutonium body burden calculations on the part of the evaluator.

INTRODUCTION

For a number of years there has been a serious need of some relatively simple method of estimating personnel body burdens of radioisotopes from urinalysis results. In recent years adequate methods have been reported[1,2] for certain specific isotopes, such as H^3 and U^{235}; however, the only known methods for determining plutonium body burden have required individual treatment combined with long experience and subjective consideration by the evaluator.

This report describes a specialized procedure for plutonium developed at this Laboratory which is not subject to the drawbacks inherent in earlier procedures. Called "PUQFUA" (PU, plutonium; Q, body burden; F, from; U, urine; A, Assays), it consists in an IBM-704 program which calculates the body burden in microcuries and the per cent of the maximum permissible body burden. Although it was written primarily for Pu^{239}, this being the only commonly encountered plutonium isotope, it is applicable also to Pu^{238} and Pu^{240}, since the permissible body burden and body

---

1. James N. P. Lawrence, Los Alamos Scientific Laboratory Report LA-2163, November 15, 1957.
2. S. R. Bernard and E. G. Struxness, Oak Ridge National Laboratory Report ORNL-2304, June 18, 1957; S. R. Bernard et al., Oak Ridge National Laboratory Report ORNL-2364, Jan. 10, 1958.

chemistry are the same for all three. As given in NBS Handbook No. 69,[3] the permissible body burden is 0.04 μc, based on the critical organ being the bone. It is assumed for PUQFUA that all personnel exposures to plutonium are acute exposures occurring at a known time.

MATHEMATICAL BASIS

PUQFUA is based on an article entitled, "The Application of Excretion Analyses to the Determination of Body Burden of Radioactive Isotopes," by Wright H. Langham.[4] Based on experimental evidence with humans, Langham developed a set of power function elimination equations for the excretion of plutonium over a five year period. These equations are:

$$Y_u = 0.002t^{-0.74} \qquad (1)$$

$$Y_{u+f} = 0.0079t^{-0.94} \qquad (2)$$

where Y_u and Y_{u+f} are the fractions of the injected dose of plutonium excreted per day in the urine and in the urine plus faeces, respectively, and t is the time in days after injection. Langham emphasizes that the errors in the constants of the above expressions may be of the order of 10 per cent.

---

3. National Committee on Radiation and Protection, NBS Handbook 69, U. S. Government Printing Office, Washington 25, D. C., June 5, 1959.
4. Wright H. Langham, British Journal of Radiology, Supplement 7, Part V, p. 95, 1957.

Since body elimination is by both urinary and faecal excretion, an integration of the expression Y_{u+f} from 1/2* to x + 1/2 days will give total fraction of the acute body burden which has been eliminated in x days (x=t). Subtracting this value from unity will give the fraction of the body burden retained (R_t) at x days after exposure.

The fraction of the original body burden (D_E) eliminated on a given day t, i.e., Y_u, will be equal to the amount of plutonium in the 24 hr urine on that day t divided by the original body burden, or

$$Y_u = \frac{U}{D_E} \tag{3}$$

where U is the amount of plutonium excreted on day t in the same unit as D_E is expressed. Combining the two expressions for Y_u, we obtain

$$D_E = 500.\ Ut^{0.74} \tag{4}$$

Thus, by measuring the 24 hr urinary excretion on any day t, we are able to compute the body burden from a single acute exposure.

In order to compute the additional body burden after another exposure, we now extend Langham's development. We calculate the urine sample to be expected from the original exposure on day t', where t' is greater than t. We then subtract this calculated daily urinary excretion from

*Arbitrarily chosen as a lower limit of integration since the power function is divergent for small values of t.

the measured value on day t' and compute D_E, using this difference as the value of U.

For successive exposures, the sum of the expected 24 hr urinary excretions is subtracted from the measured value and this difference is used in equation 4 as U to give the additional incremental body burden at the time of exposure.

By manipulating the equations previously given, the 24 hr urine specimen at some later time, t', is given by

$$U' = 0.002 D_E \, t'^{-0.74} \qquad (5)$$

Thus, by a series of successive calculations of D_E, the expected partial 24 hr urinary excretion corresponding to each urine sample may be calculated.

Once all of the partial D_E's are calculated for all the urine specimens listed, the amount of plutonium retained by the body in each case is given by

$$D_R = D_E R_t = D_E \left(1 - 0.0079 \int_{1/2}^{x+1/2} t^{-0.94} dt\right) \qquad (6)$$

or

$$D_R = D_E \left[1 - 0.1317(x + 1/2)^{0.06} + 0.1317 \, (1/2)^{0.06}\right] \qquad (6a)$$

where x is the number of days between the date of calculation and the date each partial body burden was received. Equation 6a accounts for the elimination of plutonium from the date of the exposure to the date on which the computation of body burden is made.

Hence, the total body burden on the date of calculation is given by the sum of all D_R's.

PROBLEMS IN APPLICATION

Although the theory outlined above is relatively simple, there are a number of problems which come up in the actual application. Primarily, the problem of calculating the partial body burdens from more than two or three urine specimens is a lengthy process. In order to devise an applicable system, it was decided to code the problem for the IBM-704. As presently coded, the 704 can handle the data for 500 individual urinary excretions per man. At the present rate of sampling, this capacity should be sufficient for some 50 years of data collection.

Another problem concerns the validity of plutonium urinalyses. Since about January, 1957, the method of analysis at LASL is believed to be satisfactory.[5] However, up until about the middle of 1958 there was a possibility of contamination of the specimens submitted, arising from the re-use of the metal buckets for the collection of the glass sample bottles. For these reasons, it seemed desirable to initiate a system of validation of all urinalyses before calculations of body burden were

5. Jean McClellan et al., Los Alamos Scientific Laboratory Report LA-1858 (2nd edition), p.155, August 1958.

performed. This validation is accomplished in the following manner. Each successive sample is used to validate the preceding sample. Based purely on the magnitude of the preceding sample, the expected urinary excretion on the date of the immediate successive sample is calculated using equation 5. If the successive sample (in d/m) is equal to, or greater than, the calculated value (in d/m) minus 0.1 d/m, then the preceding sample is considered valid. If the successive sample result is less than this number, the preceding sample is set equal to zero for purposes of calculation.

At first glance this method may seem arbitrary. However, it should be realized that the several samples preceding the one to be validated also may indicate an incremental body burden, part of which is constantly being eliminated. The effect of these incremental body burdens will tend to increase the measured excretion rate over that predicted by only the immediately preceding sample. Thus, the only possible flaw would be validation of samples which do not add to the total body burden, and these will be eliminated in the complete calculation. This technique automatically makes the latest urine sample of a series valid (i.e., until another sample is submitted).

Another minor difficulty was experienced in supplying the IBM-704 with dates of sampling in terms suitable for calculation. This problem was overcome by establishing an initial date of January 1, 1944, and having the 704 compute the time interval in days from this date for all dates appearing in the input data.

Since this system assumes all exposures to be acute, the date of actual exposure is required in the calculation. If a known accident occurred, then this date is directly fed into the calculation. If no known accident occurred, then it is assumed that the exposure occurred on the date half-way between two successive submissions of urine samples. At LASL, plutonium urine sampling is done on approximately a monthly basis for the persons most likely to be exposed. Thus, in the majority of cases, the exposure would be assumed to occur about fifteen days before the sample was taken. Therefore, in coding quite arbitrarily the date of exposure for the first sample submitted is assumed to be 15 days before the sample submission date, unless a known accident occurred, and then the actual date is used.

In certain cases following an accident, treatment with EDTA or other drugs is administered to the person involved. Urine samples submitted during the treatment and for a period after the treatment has ceased are omitted from the input data. In such cases, the urinary excretion rate is substantially higher than that predicted by the Langham equations, and such urine samples should not be included in the calculations.

For convenience, all primary calculations of body burden are done in terms of disintegrations per minute. For clarity of presentation, these are converted to microcuries (μc) and fraction of permissible body burden. Also the partial body burden received during the preceding six months and during each preceding year is calculated.

704 PROGRAM

The system devised for calculation of body burden is immediately applicable at other installations which have access to IBM-704 equipment with 16 K word capacity, provided the input data are supplied in the form specified below. No claim is made as to the conciseness of the form in which the problem was originally coded. Other simpler and more economical codings are undoubtedly possible.

Primarily PUQFUA is a FORTRAN coding. However, in order to convert months expressed as numbers from 1 through 12 to numbers from 1 through 9 plus letters O, N, D, a short 871 compilation was required, and the basic FORTRAN program was slightly modified to permit inclusion of this sub program. Incorporated in the basic program are instructions to write on peripheral tape equipment the data necessary to punch the required information in an INTERNAL EXPOSURE RECORD CARD, shown below.



The FORTRAN statements, which comprise the program, follow:

```
 DIMENSION ID(500),IDUS(500),EDUS(500),US(500),FEUS(500),UUS(500),
XIDPQ(500),PQ(500),EUS(500),EXC(500),CPQ(500),ICD(1),CQ(1),FCQ(1),
XCCQ(1),IDATE(24),ELCPQ(500),ELCQ(24),FELCQ(24),AFCQ(24),
XIDELMO(24),IDELDA(24),IDELYR(24),KMO(500),KYR(500),KDA(500),
XLYR(24),LMO(13),LLMO(13),IDMO(500),IDDA(500),IDYR(500),DUMMY(12)
 REWIND8
 REWIND2
 REWIND3
 READINPUTTAPE8,4,(LYR(J),J=1,24)
 READINPUTTAPE8,4,(LMO(J),J=1,13)
 READINPUTTAPE8,4,(LLMO(J),J=1,13)
 READINPUTTAPE8,6,(IDELMO(J),IDELDA(J),IDELYR(J),J=1,24)
 DO650J=1,24
 LL=IDELYR(J)-43
 IF(LL)664,664,665
665 IYR=LYR(LL)
 IF(LL-1)1001,661,651
651 IF(LL-5)662,661,652
652 IF(LL-9)662,661,653
653 IF(LL-13)662,661,654
654 IF(LL-17)662,661,655
655 IF(LL-21)662,661,656
656 IF(LL-25)662,661,1001
```

```
661 KM=IDELMO(J)
    IMO=LLMO(KM)
    GOTO663
662 KM=IDELMO(J)
    IMO=IMO(KM)
663 IDATE(J)=IYR+IMO+IDELDA(J)
    GOTO650
664 IDATE(J)=0
650 CONTINUE
    READINPUTTAPE8,3,NNN
110 READINPUTTAPE8,3,N,ICDMO,ICDDA,ICDYR
    READINPUTTAPE8,4,(ID(J),J=1,N)
    READINPUTTAPE8,6,(KMO(J),KDA(J),KYR(J),J=1,N)
    READINPUTTAPE8,5,(US(J),J=1,N)
    LL=ICDYR-43
    IYR=LYR(LL)
    IF(LL-1)1001,681,682
682 IF(LL-5)680,681,683
683 IF(LL-9)680,681,684
684 IF(LL-13)680,681,685
685 IF(LL-17)680,681,686
686 IF(LL-21)680,681,687
687 IF(LL-25)680,681,1001
681 IMO=LLMO(ICDMO)
```

```
      GOTO688
  680 IMO=LMO(ICDMO)
  688 ICD=IYR+IMO+ICDDA
      DO750J=1,N
      LL=KYR(J)-43
      IYR=LYR(LL)
      IF(LL-1)1001,761,751
  751 IF(LL-5)762,761,752
  752 IF(LL-9)762,761,753
  753 IF(LL-13)762,761,754
  754 IF(LL-17)762,761,755
  755 IF(LL-21)762,761,756
  756 IF(LL-25)762,761,1001
  761 KM=KMO(J)
      IMO=LLMO(KM)
      GOTO763
  762 KM=KMO(J)
      IMO=LMO(KM)
  763 IDUS(J)=IYR+IMO+KDA(J)
  750 CONTINUE
      EDUS(2)=FLOATF(IDUS(2)-IDUS(1)+15)
      FEUS(2)=US(1)*(15.**0.74)*1./EDUS(2)**0.74
      IF(US(2)-FEUS(2)+0.1)300,301,301
  300 UUS(1)=0.
```

```
      GOTO400
  301 UUS(1)=US(1)
  400 DO401J=3,N
      EDUS(J)=FLOATF(2*IDUS(J)-IDUS(J-1)-IDUS(J-2))/2.
      FEUS(J)=US(J-1)*((FLOATF(IDUS(J-1)-IDUS(J-2))/2.)**0.74)*
     X1./EDUS(J)**0.74
      IF(US(J)-FEUS(J)+0.1)302,303,303
  302 UUS(J-1)=0.
      GOTO401
  303 UUS(J-1)=US(J-1)
  401 CONTINUE
  501 UUS(N)=US(N)
      IDPQ(1)=IDUS(1)-ID(1)
      PQ(1)=500.*UUS(1)*(FLOATF(ID(1))**0.74
      DO60L=2,N
      M=L-1
      DO30K=1,M
      EUS(K)=0.002*PQ(K)*1./(FLOATF(IDUS(L)-IDPQ(K))**0.74)
   30 CONTINUE
      SEUS=0.
      DO33K=1,M
   33 SEUS=SEUS+EUS(K)
      EXC(L)=UUS(L)-SEUS
   34 IF(EXC(L)    )35,40,40
```

```
 35 IDPQ(L)=0
    PQ(L)=0.
    GOTO60
 40 IF(ID(L))47,47,50
 50 IDPQ(L)=IDUS(L)-ID(L)
    PQ(L)=500.*EXC(L)*(FLOATF(ID(L))**0.74)
 46 GOTO60
 47 CD=FLOATF(IDUS(L)-IDUS(L-1))/2.
    IDPQ(L)=IDUS(L)-XFIXF(CD)
    PQ(L)=500.*EXC(L)*(CD**0.74)
 60 CONTINUE
    DO80L=1,N
    CNST=1.+0.1317*(0.5**0.06)
 80 CPQ(L)=PQ(L)*(CNST-0.1317*((FLOATF(ICD-IDPQ(L))+0.5)**0.06))
    CQ=0.
    DO100L=1,N
100 CQ=CQ+CPQ(L)
    FCQ=CQ/97680.
    CCQ=FCQ*0.044
    DO204K=1,24
    DO200J=1,N
    IF(IDPQ(J)-IDATE(K))201,201,202
201 ELCPQ(J)=0
    GOTO200
```

```
202 ELCPQ(J)=CPQ(J)
200 CONTINUE
    ELCQ(K)=0.
    DO203J=1,N
203 ELCQ(K)=ELCQ(K)+ELCPQ(J)
    FELCQ(K)=ELCQ(K)/97680.
204 CONTINUE
    AFCQ(1)=FELCQ(1)
    AFCQ(2)=FELCQ(2)
    DO205K=3,24
205 AFCQ(K)=FELCQ(K)-FELCQ(K-1)
    CAFCQ=FELCQ(1)*0.044
    DO940J=1,N
    IF(IDPQ(J))1001,948,949
948 IDMO(J)=0
    IDDA(J)=0
    IDYR(J)=0
    GOTO940
949 K=1
952 K=K+1
    III=IDPQ(J)-LYR(K)
    KKK=IDPQ(J)-LYR(K-1)
    IF(III)950,951,952
951 IDYR(J)=K+42
```

```
      IDMO(J)=12
      IDDA(J)=31
      GOTO940
  950 IDYR(J)=K+42
      IF(K-2)1001,961,953
  953 IF(K-6)962,961,954
  954 IF(K-10)962,961,955
  955 IF(K-14)962,961,956
  956 IF(K-18)962,961,957
  957 IF(K-22)962,961,1001
  961 L=1
  965 L=L+1
      LLL=KKK-LLMO(L)
      JJJ=KKK-LLMO(L-1)
      IF(LLL)964,964,965
  964 IDMO(J)=L-1
      IDDA(J)=JJJ
      GOTO940
  962 L=1
  967 L=L+1
      LLL=KKK-IMO(L)
      JJJ=KKK-IMO(L-1)
      IF(LLL)966,966,967
  966 IDMO(J)=L-1
```

```
      IDDA(J)=JJJ
  940 CONTINUE
      IFCQ=FCQ*1000.
      ICCQ=CCQ*10000.
      IAFCQ=AFCQ(1)*1000.
      ICAFCQ=CAFCQ*10000.
      KKMO=KMO(N)
  801 READINPUTTAPE8,2,IH,IC,IFIC,ICFIC,ISMFIC,ISMCFI,IA,IG,NUM
      DIOT=DUMMY(KKMO)
      WRITEOUTPUTTAPE3,2,ICDMO,ICDYR,IFCQ,ICCQ,IAFCQ,ICAFCQ,
     XKDA(N),KYR(N),NUM
      WRITEOUTPUTTAPE2,1
      WRITEOUTPUTTAPE2,2,ICDMO,ICDYR,IFCQ,ICCQ,IAFCQ,ICAFCQ,
     XKDA(N),KYR(N),NUM
      IF(SENSESWITCH1)111,112
  111 PRINT2
  112 IF(SENSESWITCH2)502,503
  502 WRITEOUTPUTTAPE2,13
      WRITEOUTPUTTAPE2,10,(IDMO(J),IDDA(J),IDYR(J),CPQ(J),KMO(J),
     XKDA(J),KYR(J),US(J),UUS(J),J=1,N)
  503 WRITEOUTPUTTAPE2,15
      WRITEOUTPUTTAPE2,16,FCQ,(AFCQ(K),K=1,11)
      WRITEOUTPUTTAPE2,17
      WRITEOUTPUTTAPE2,16,(AFCQ(K),K=12,24)
```

```
      IF(NNN-NUM)810,810,110
  810 ENDFILE2
      ENDFILE3
      REWIND2
      REWIND3
      REWIND8
      STOP77777
 1001 PRINT7
      STOP77777
    1 FORMAT(73H                                          FCQ MCCQ AFCQ CAFCQ
     XDATE LAST SAMPLE)
    2 FORMAT(29H                                 I1,I2,4I5,
     X11H              I2,I2,9H     I4)
    3 FORMAT(I12,I8,2I2)
    4 FORMAT(12I6)
    5 FORMAT(7F10.4)
    6 FORMAT(12(3I2))
    7 FORMAT(15H MACHINE ERROR   )
   10 FORMAT(I33,2I5,E11.4,3I5,2E11.4)
   13 FORMAT(90H                                          IDMO IDDA IDYR            CPQ
     X   KMO    KDA    KYR     US              UUS   )
   15 FORMAT(97H     FCQ   1/2 YEAR 1 YEAR 2 YEAR 3 YEAR 4 YEAR 5 YEAR
     X   6 YEAR   7 YEAR   8 YEAR   9 YEAR   10 YEAR   )
   16 FORMAT(15F8.4)
```

```
17 FORMAT(105H  11 YEAR  12  YEAR  13  YEAR  14  YEAR   15  YEAR  16  YEAR   17  YEA
  XR   18  YEAR  19  YEAR  20  YEAR  21  YEAR  22  YEAR  23  YEAR  )
```

In order of their appearance in the listing above a short description follows of each of the variables used.

ID	Number of days before urine sample submission date on which accident occurred; if no known accident, zeros are used, except for the first sample when 15 is used.
IDUS	Date of submission of urine sample converted to days since 1/1/44 (machine date)
EDUS	Elapsed days from exposure for immediately previous urine sample to date of current sample
US	Urine assay data in d/m - 24 hr sample
FEUS	Estimated urine sample (in d/m - 24 hr sample) from immediately previous sample
UUS	Validated urine assay data in d/m - 24 hr sample
IDPQ	Date of exposure for current urine sample (machine date)
PQ	Partial body burden (in d/m) at time of exposure (IDPQ)
EUS	Estimated partial urine sample based on PQ
EXC	Actual urine sample minus sum of EUS's
CPQ	Partial body burden (in d/m) due to PQ at date of calculation
ICD	Date of calculation (machine date)
CQ	Sum of CPQ's (or total body burden) in d/m
FCQ	Total body burden as fraction of permissible
CCQ	Total body burden in microcuries
IDATE	Machine date for determining partial exposure which occurred in successive years
ELCPQ	CPQ occurring between IDATE and ICD

ELCQ	Sum of ELCPQ's for successive years in d/m
FELCQ	Fraction of body burden acquired between IDATE and ICD
AFCQ	Difference in two successive values of FELCQ, or fraction of body burden acquired in successive years
IDELMO, IDELDA, IDELYR	Month, day, and year (2 digit notation) of previous years to date of calculation
KMO, KYR, KDA	Month, year, and day of submission of urine sample
LYR	Number of days from 1/1/44 to the first of each successive year
LMO	Number of days from first day of any non-leap year to first day of each month
LLMO	Number of days from first day of any leap year to first day of each month
IDMO, IDDA, IDYR	Month, day, and year date of exposure to plutonium, IDPQ converted to month, day, and year
DUMMY	A set of twelve constants which are compiled by an 871 assembly, and which alter the FORTRAN assembly as mentioned earlier
LL	Calculation constant defined variously in FORTRAN statements
IYR	Partial date used to compute IDATE or ICD
KM	Calculation constant defined variously in FORTRAN statements
IMO	Partial date used to compute IDATE or ICD
NNN	Total number of persons for whom body burden calculations are to be made

N	Total number of urine samples to be used for a single calculation of body burden
ICDMO ICDDA ICDYR	Month, day, and year date of calculation
M	Calculation constant defined in FORTRAN statements
SEUS	Sum of EUS's
CD	Half the number of days between successive urine samples
CNST	Constant defined in FORTRAN statements
III KKK LLL JJJ	Calculation constants variously defined in FORTRAN statements
IFCQ ICCQ IAFCQ ICAFCQ	FCQ,CCQ,AFCQ,CAFCQ multiplied by a constant and truncated, suitable for punching on INTERNAL EXPOSURE RECORD cards
KKMO	Month of submission of last urine sample in current calculation
IH IC IFIC ICFIC	Dummy functions required to fill spaces in input statements

ISMFIC
ISMCFI
IA
IG

} Dummy functions required to fill spaces in input statements

NUM — Four digit identification for each person whose body burden is calculated. No number should exceed NNN and the last person to be calculated should have NUM=NNN

DIOT — Causes one bit of FORMAT to be changed according to value of DUMMY selected

As mentioned earlier, an 871 assembly was required. After assembling PUQFUA by FORTRAN, the location of the storage for DUMMY must be obtained. In the original compilation, this occurred from 23545 through 23534. The following deck, assembled by 871, provides the data to be stored as DUMMY.

```
5006      PUQFUA
188000000      00+23545
H880001J0                D12,1
H880001-0                N12,1
H880000R0                012,1
H880000Q0                912,1
H880000P0                812,1
H88000000                712,1
H880000N0                612,1
H880000M0                512,1
H880000L0                412,1
```

H880000K0 312,1

H880000J0 212,1

H88000000 112,1

In addition, the binary deck of PUQFUA must be altered by changing the address of the instruction STO DIOT from the address of DIOT to the address of the 13th bit of the assembled FORMAT statement #2.

With the modifications as described, by using peripheral equipment, a tape-controlled printer, and a tape-to-card converter, PUQFUA can supply the following:

(1) INTERNAL EXPOSURE RECORD cards for each individual.



(2) Listings containing the identification of the person, FCQ, CCQ, AFCQ, CAFCQ, date of last sample and the person's identification number; the fractional body burden acquired in the last half year, the last year, the next to last year (2nd year), third year, etc., from the

date of calculation.

FCQ MCCQ AFCQ CAFCQ DATE LAST SAMPLE

DOE J S 00002H0129758001070048 0 0 82257PLUTONIUM 267

FCQ	1/2 YEAR	1 YEAR	2 YEAR	3 YEAR	4 YEAR	5 YEAR	6 YEAR	7 YEAR
0.1080	0.	0.	0.	0.	0.0324	0.0077	0.0596	0.0084

8 YEAR	9 YEAR	10 YEAR	11 YEAR	12 YEAR	13 YEAR	14 YEAR
0.	0.	0.	0.	0.	0.	0.

(3) If SENSE SWITCH 2 is depressed, in addition to the listing described in (2) above, the individual dates of assumed exposure, the calculated exposure in disintegrations per minute, the date of each urine sample, the value of each urine sample, and the corresponding validated urine samples.

FCQ MCCQ AFCQ CAFCQ DATE LAST SAMPLE

DOE J S 00000H0129758001080000480 0 0 82257 PLUTONIUM 267

IDMO	IDDA	IDYR	CPQ	KMO	KDA	KYR	US	UUS
3	30	52	0.8162E 03	4	14	52	0.2400E-00	0.2400E-00
7	31	52	0.3335E 04	11	15	52	0.2600E-00	0.2600E-00
4	30	53	0.2485E 04	10	13	53	0.2200E-00	0.2200E-00
3	20	54	0.7489E 03	8	25	54	0.1600E-00	0.1600E-00
2	20	55	0.3164E 04	8	18	55	0.2500E-00	0.2500E-00
1	0	44	0.	8	21	56	0.1200E-01	0.1200E-01
1	0	44	0.	8	22	57	0.	0.

FCQ	1/2 YEAR	1 YEAR	2 YEAR	3 YEAR	4 YEAR	5 YEAR	6 YEAR	7 YEAR
0.1080	. 0.	0.	0.	0.	0.0324	0.0077	0.0596	0.0084

8 YEAR	9 YEAR	10 YEAR	11 YEAR	12 YEAR	13 YEAR	14 YEAR	15 YEAR
0.	0.	0.	0.	0.	0.	0.	0.

INPUT DATA

In order to minimize the time required for entry of the input data, PUQFUA was written to accept all input data from tapes and not cards. All input data is first written from cards on Tape 8 by means of peripheral equipment. The order of the input data can be obtained from the FORTRAN statements; however, a short description of each block of data follows.

The first deck of input data consists of the LYR table. In this table are the 24 numbers corresponding to the total number of days at the end of 24 consecutive years starting January 1, 1944. This deck must be punched by FORMAT (12I6).

The second deck is the IMO table. This table consists of the number of days at the beginning of each month as counted from the first day of non-leap years, punched by FORMAT (12I6). The table contains thirteen entries, the last being the number of days in the year.

The third deck is the LIMO table. This table is the leap year table corresponding to the IMO table.

The fourth deck is comprised of dates. The first date is that six months prior to the calculation date. Thereafter, the dates are those for each year through the twenty-third year prior to the date of the calculation. A set of three two-digit numbers comprises the date, with the first being the month, the second the day, and the last the year. These dates are arranged according to FORMAT (12(3I2)).

The fifth deck consists of the single number NNN punched according to FORMAT (I12, I8, 2I2).

The sixth deck contains all the remaining necessary information for computing body burden. It is comprised of NNN sets of five subdecks, where the number NNN is the total number of persons whose body burdens are to be evaluated.

Subdeck 1 (or the NCDL deck) consists of a single card with the numbers N, ICDMO, ICDDA, ICDYR punched according to FORMAT (I12, I8, 2I2). Although not called for by the FORMAT, in columns 73 through 76 are the letters NCDL, and in columns 77 through 80 the identification number of the person (NUM).

Subdeck 2 (or the ID deck) consists of a table of N entries corresponding to the various ID's of each urine sample, punched according to FORMAT (12I6). While not called for by the FORMAT, in columns 73 and 74 are the letters ID and in columns 77 through 80, the corresponding NUM.

Subdeck 3 (or DU deck) consists of the dates of the N urine samples punched according to FORMAT (12(3I2)). Again not called for by the FORMAT, in columns 73 and 74 are the letters DU and in columns 77 through 80 the corresponding NUM.

Subdeck 4 (or US deck) consists of N actual urine assay results in d/m - 24 hr sample for the person concerned, punched according to FORMAT (7F10.4). Again not called for by the FORMAT, in columns 73 and 74 are the letters US, and in columns 77 through 80 the corresponding NUM.

Subdeck 5 (or PLUTONIUM deck) is a single card deck containing the identification of the individual. The information is punched by the following FORMAT:

FORMAT (29H   I1, I2, 4I5, 11H   I2, I2
9H   I4):

Column 1, blank

Columns 2 through 19, last name and the initials of person being evaluated

Columns 20 through 24, Z number (LASL identification number) of person being evaluated

Columns 25 through 27, current LASL group code

Columns 28 and 29, last digits of the birth year of person being evaluated

Columns 30 through 67, no punches

Columns 68 through 76, the word PLUTONIUM

Columns 77 through 80, NUM of person being evaluated

DISCUSSION OF RESULTS

PUQFUA purports to measure the body burden exclusive of the lung burden. Since it is based on urinalysis data, only that plutonium which has been absorbed into the system influences the calculations. It has been found that the body burden of certain persons appears to increase continuously at a slow rate, even though these persons are no longer exposed to plutonium. This is believed to be due to a slow absorption of a suspected plutonium lung burden into the blood stream. Thus, for all intents and purposes, persons with significant lung burdens have constant sources of exposure to plutonium.

Any system of computing body burden is only as reliable as the data which go into the calculation. When Langham estimated the body burdens of the cases in his paper, he had his own system of validating urine samples based on his past experience and extensive knowledge of the subject. The validation method of PUQFUA requires no past experience on the part of the person performing the calculation and is objective in treatment of data.

PUQFUA calculations were compared with Langham's original calculations, and the listing below gives Langham's estimated body burden (converted to microcuries) and the body burdens as calculated by PUQFUA. Positive exposures in these cases occurred during the period 1944-46. On the average the body burdens computed by PUQFUA are about 1.6 times those of Langham.

Based on tissue analysis of a recently deceased Los Alamos employee, his total body burden was found to be 0.0175 μc.* Because of the small tissue samples taken and the assumption of uniform distribution throughout each organ tissue, there is an undetermined uncertainty in the body burden figure. PUQFUA calculation on the same man, using urinanalysis records, gives body burden of 0.0191 μc.

In view of the wide spread of the data entering the urinalysis

*Private communication with H. Foreman and W. H. Langham - Group H-4, LASL

Case code	Langham's body burden μc(±50%)	PUQFUA body burden μc(±50%)
W. G.	0.08	0.12
W. B.	0.07	0.13
D. D.	0.07	0.14
D. W.	0.06	0.07
W. A.	0.06	0.11
G. F.	0.06	0.08
R.D.B.	0.05	0.03
F. C.	0.04	0.06
H. R.	0.04	0.09
W. S.	0.04	0.04
T. M.	0.03-0.06	0.01
H. L.	0.03	0.06
T. E.	0.02	0.03
R.A.B.	0.02	0.02
M. W.	0.02	0.03
D. K.	0.02	0.03
D. H.	0.02	0.02
K. E.	0.02	0.03
J. C.	0.02	0.04
J. B.	0.02	0.04
J. A.	0.02	0.03
E. R.	0.01	0.04
C. H.	0.01	0.02
J. O.	0.006	0.01
C. D.	0.006	0.002
A. B.	0.006	0.01

calculation prior to the adoption of the nuclear track plate technique[5] and the uncertainties in the tissue sample calculation, PUQFUA is in excellent agreement with the other methods of determining plutonium body burden. Also since the adoption of the nuclear track plate method of plutonium urinalysis, the results have been far more consistent. As a consequence, practically all samples are validated. When all data entering the calculations are of this type, even closer agreement is anticipated between the various methods of computing plutonium body burden based on urine assays.